

    

        KinoExRatatui

        v0.2.2



    


  

    Table of contents

    
      



      	Overview


      	Contributing


      	Changelog





    	Guides
      


      	Configuration


      	Telemetry



      

    




        	
          Modules
          

    	Widgets
      


      	Kino.ExRatatui



      

    




    	Internals
      


      	Kino.ExRatatui.Telemetry



      

    




        



      

    

  

    KinoExRatatui

Run ExRatatui apps inside Livebook notebooks.
[image: KinoExRatatui Demo]
KinoExRatatui is a byte-stream transport that pipes the runtime's rendered ANSI through xterm.js and forwards keypresses and resize events back. Implemented as a Kino.JS.Live widget on top of ExRatatui.Transport.ByteStream.
Features
	Same App, same surface — any module implementing ExRatatui.App runs unchanged.
	Responsive sizing — xterm.js's FitAddon derives cell dimensions and reports resize events; the App sees them as %ExRatatui.Event.Resize{} in handle_event/2.
	Static frames — Kino.ExRatatui.frame/2 renders a one-shot [{widget, rect}, ...] list and ships the bytes to xterm.js. Useful for documentation, side-by-side comparisons via Kino.Layout.grid/1, screenshots, etc.
	Inline images — the bundled @xterm/addon-image registers Sixel and iTerm2 inline-image parsers, so ExRatatui.Widgets.Image renders PNG / JPEG / GIF / WebP / BMP end-to-end in Livebook. Build images with ExRatatui.Image.new/2 and place them in the widget tree like any other widget; pass protocol: :sixel or protocol: :iterm2 at construction time.
	Themeable — pass :theme, :font_family, :font_size, :height, :cursor_blink, :scrollback, and :stopped_message to new/2 (or the static-friendly subset to frame/2) to override the defaults per cell. The :theme map is the full xterm.js ITheme — 16 ANSI colors, selection, cursor accents, the lot. Use the :dark / :light / :livebook atom shorthands to pick a bundled palette; :livebook follows the user's prefers-color-scheme and live-switches.
	Global defaults — Kino.ExRatatui.configure/1 writes display defaults to the :kino_ex_ratatui Application environment. Per-instance opts still win key-by-key. See the Configuration guide.
	Accessible stopped state — when the runtime exits the widget renders a role="status" aria-live="polite" DOM overlay over the xterm container. Screen readers announce it; sighted users see a clean italic message instead of the frozen final frame. Customise the text with :stopped_message.
	Zero browser-side state on cell re-eval — re-running the cell tears the runtime down and starts a fresh one, matching every other Kino.JS.Live widget.
	Telemetry — [:kino_ex_ratatui, :transport, :connect | :disconnect], [:kino_ex_ratatui, :render, :frame], [:kino_ex_ratatui, :input, :forward], and [:kino_ex_ratatui, :resize] events sit one layer above ex_ratatui's own runtime/render telemetry. See the Telemetry guide for the full event catalogue and a Telemetry.Metrics example.


Examples
Five notebook examples live under examples/ — open them in Livebook and run the cells. See the catalog for a one-liner per notebook and a recommended starting point.
Ecosystem
	ex_ratatui — The core terminal UI library this builds on.
	phoenix_ex_ratatui — Run TUIs in the browser within Phoenix LiveView.

Installation
Add kino_ex_ratatui to the Livebook setup cell (or the project's mix.exs):
Mix.install([
  {:kino_ex_ratatui, "~> 0.2"}
])
Prerequisites
	Elixir 1.17+
	Livebook 0.19+

Quick Start
defmodule Counter do
  use ExRatatui.App

  alias ExRatatui.Event.Key
  alias ExRatatui.Layout.Rect
  alias ExRatatui.Widgets.{Block, Paragraph}

  def mount(_), do: {:ok, %{n: 0}}

  def render(state, frame) do
    [
      {%Paragraph{
         text: "Count: #{state.n}\n\n+ increment   - decrement   q quit",
         block: %Block{title: "counter"}
       },
       %Rect{x: 0, y: 0, width: frame.width, height: frame.height}}
    ]
  end

  def handle_event(%Key{code: "+"}, s), do: {:noreply, %{s | n: s.n + 1}}
  def handle_event(%Key{code: "-"}, s), do: {:noreply, %{s | n: s.n - 1}}
  def handle_event(%Key{code: "q"}, s), do: {:stop, s}
  def handle_event(_, s),                do: {:noreply, s}
end

Kino.ExRatatui.new(Counter)
Static frames
alias ExRatatui.Layout.Rect
alias ExRatatui.Widgets.{Block, Paragraph}

Kino.ExRatatui.frame(
  [
    {%Paragraph{
       text: "Hello from a static frame!",
       block: %Block{title: "demo"}
     },
     %Rect{x: 0, y: 0, width: 40, height: 5}}
  ],
  cols: 40,
  rows: 5
)
frame/2 renders the widget list once via ExRatatui.Session, ships the resulting ANSI to xterm.js, and stops. No event loop, no runtime server.
How it works
KinoExRatatui implements ExRatatui.Transport as a byte-stream transport — the same shape as the built-in SSH transport. The wiring:
xterm.js (iframe)            Kino.ExRatatui (Kino.JS.Live)         ExRatatui.Server
─────────────────            ─────────────────────────────         ────────────────
onData(bytes)         ──>    handle_event("input", _)        ──>   {:ex_ratatui_event, _}
ResizeObserver        ──>    handle_event("resize", _)       ──>   {:ex_ratatui_resize, _, _}
xterm.write(bytes)    <──    broadcast_event("ansi", _)      <──   writer_fn.(bytes)
The runtime server starts lazily on the first "resize" event so the ExRatatui.Session opens at the exact dimensions xterm.js's FitAddon settled on. From there, input bytes round-trip through ExRatatui.Transport.ByteStream.forward_input/3 (which absorbs synthesized Event.Resize events and dispatches everything else as {:ex_ratatui_event, _}). When the App returns {:stop, _}, the live widget catches the runtime's :DOWN and broadcasts a stop state message.
Guides
	Guide	Description
	Configuration	Global display defaults via configure/1, the theme atom shorthands, and the merge order
	Telemetry	:telemetry events for transport, render, input, and resize — logging and Telemetry.Metrics

Contributing
See CONTRIBUTING.md for development setup and guidelines.
KinoExRatatui is built on ExRatatui, a general-purpose terminal UI library for Elixir. Contributions to its underlying rendering, widgets, or layout engine are very welcome too.
License
MIT — see LICENSE.


  

    Contributing to KinoExRatatui

Thanks for the interest in contributing!
KinoExRatatui is built on ExRatatui. Consider contributing to the upstream library too when a feature is missing or something is not working. Contributions are welcome everywhere!
This guide covers the setup.
Setup
	Clone the repo:

git clone https://github.com/mcass19/kino_ex_ratatui.git
cd kino_ex_ratatui

	Prerequisites:

	Elixir 1.17+ and Erlang/OTP 26+.
	Node.js 20+ — only needed to rebuild the xterm.js bundle. End users installing from hex don't need it.

	Fetch dependencies:

mix deps.get
mix assets.install   # cd assets && npm install

Running Tests
mix test
mix test --cover        # must report 100.00% Total

The suite uses Kino.Test's configure_livebook_bridge setup to drive the live widget end-to-end without a real browser. It includes property-based invariants via stream_data — covering display-option round-trip, mount-opts strip-out, ordering preservation, and configure/1 precedence. Properties run as part of the regular mix test invocation.
For the actual browser smoke test, open one of the notebooks under examples/ in Livebook and run the cells.
Bundling the JS
mix assets.build       # cd assets && npm run build (minified)
# or
cd assets && npm run build:dev   # with sourcemaps

The bundled output lands at lib/assets/kino_ex_ratatui/main.js (and main.css for xterm's stylesheet). Both files are committed so the published hex package needs no Node toolchain at install time.
When anything under assets/js/ changes, rerun mix assets.build and commit the regenerated bundle.
The hook's pure logic (theme/display resolution) has unit tests under assets/test/. Run them with cd assets && npm test — Node's built-in test runner, no extra deps. CI runs them on every push.
Branching and Commits
	Branch from main
	Keep commits focused and atomic
	Use descriptive commit message prefixes: feat:, fix:, docs:, test:, refactor:, chore:

Pull Requests
Before submitting a PR, make sure the following pass:
mix format --check-formatted
mix compile --warnings-as-errors
mix credo --strict
mix dialyzer
mix test --cover

	Keep PRs focused — one feature or fix per PR
	Add tests for new functionality
	Add @doc, @spec, and @moduledoc for new public functions and modules
	Update documentation (moduledocs, CHANGELOG, README if applicable)
	For breaking changes, include migration notes in the CHANGELOG
	Follow existing code style and patterns
	Ensure CI passes before requesting review



  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
Unreleased
0.2.2 - 2026-06-03
Fixed
	Correct the documented Livebook floor to 0.19+ (it tracks the kino ~> 0.19 requirement), and repair the malformed [0.1.1]–[0.2.1] CHANGELOG compare links.

Changed
	Relax the telemetry dependency to ~> 1.0, matching ex_ratatui and phoenix_ex_ratatui.
	Split the pure theme/display resolution (resolveTheme, resolveDisplay, prefersDark, and the bundled theme/default constants) out of the JS hook into a dependency-free assets/js/display.js, and cover it with node --test unit tests (assets/test/display.test.js). main.js imports it; the bundled lib/assets/kino_ex_ratatui/main.js is byte-different but behavior-identical.
	Move the README demo GIF from assets/ to .github/demo.gif, leaving assets/ for the JS build pipeline only.
	Add an Ecosystem section and a Guides table to the README, surfacing the Configuration and Telemetry guides — matching ex_ratatui and phoenix_ex_ratatui.

Internal
	CI now installs JS deps and runs the JS hook tests as their own steps (matching phoenix_ex_ratatui), and Dependabot tracks the assets/ npm packages.

0.2.1 - 2026-05-20
Added
	Inline image rendering via @xterm/addon-image. The bundled JS hook now loads @xterm/addon-image 0.9.0 in both the live (new/2) and static (frame/2) init paths, registering Sixel (DCS q ...) and iTerm2 inline-image (ESC ] 1337 ; File=... ^G) parser handlers on the xterm.js terminal. Without it those escape sequences were silently swallowed and ExRatatui.Widgets.Image (introduced in ex_ratatui 0.10) rendered nothing in Livebook. Construct images with ExRatatui.Image.new/2 passing protocol: :sixel or protocol: :iterm2 — xterm.js does not implement the Kitty graphics protocol. Bundle grows ~60 KB minified; no Elixir API change, no per-instance opt change. Moduledoc gained an Inline images section pointing at the ex_ratatui Images guide.

	examples/new_widgets.livemd — tour of the three widgets introduced in ex_ratatui 0.10. One consolidated Livebook notebook covering ExRatatui.Widgets.Image (a static frame with the new image addon plus an interactive App that cycles :sixel / :iterm2 at runtime), ExRatatui.Widgets.CodeBlock (syntect-highlighted snippet with :solarized_dark, line numbers, and emphasis on lines 3..5), and ExRatatui.Widgets.BigText (slide-deck banner showing :full and :quadrant density variants). Closes with a Kino.Layout.grid/2 side-by-side comparison of all three. Catalogue entry added to examples/README.md. Mix.install pins ~> 0.3 so the notebook picks up the addon-image bundle as soon as 0.3.0 is published.


0.2.0 - 2026-05-07
Added
	Accessible stopped-state DOM overlay. When the runtime server exits ({:stop, _}, mount/1 failure, crash) the widget no longer paints a dim ANSI message into the xterm buffer — it broadcasts a "stopped" Kino event with %{message: stopped_message}, and the JS hook anchors a role="status" aria-live="polite" <div> over the xterm container. Screen readers announce the message via aria-live; sighted users see a centered italic line in the configured theme's foreground/background instead of a dead cursor sitting on the frozen final frame. The overlay is pointer-events: none, applied at z-index: 1, and uses textContent so a user-supplied :stopped_message is never interpreted as HTML. Idempotent — refires of the event are ignored once an overlay is present. The :stopped_message knob from the per-instance display options is unchanged; this is purely a wire-protocol + presentation switch from "ANSI bytes the user reads off the dead xterm buffer" to "structured payload an accessible DOM overlay renders".

Changed
	"stopped" Kino event replaces the in-buffer ANSI message on server :DOWN. The wire protocol changed: Kino.JS.Live.broadcast_event(ctx, "ansi", {:binary, %{}, "\\e[2J… App stopped …\\e[0m"}) is now broadcast_event(ctx, "stopped", %{message: stopped_message}). The build_stopped_screen/1 private helper is removed. Existing tests updated; the assert_broadcast_event(kino, "stopped", %{message: _}) shape pins the new contract, plus a defensive test that the payload is a plain map (not {:binary, _, _}) so a regression to byte-stream-shaped messages would fail loudly. End-user :stopped_message API is unchanged.

	Kino.ExRatatui.configure/1 — global display defaults. Writes to the :kino_ex_ratatui Application environment so every subsequent new/2 and frame/2 call picks them up without ceremony. Accepts the same seven display keys as new/2. Validation runs at the call site (matches new/2's shape) so a typo in configure/1 fails fast rather than producing a working-but-wrong widget. Calling twice merges into the prior config rather than replacing it, so related settings can split across cells. Merge order is per-instance opts > configure/1 > module defaults, key-by-key — partial config sticks. The same env is reachable via Application.get_all_env(:kino_ex_ratatui) so a release config/runtime.exs works equivalently. New Configuration guide walks through the merge order, the atom theme shorthands, and an "when not to use it" pointer for runtime-dependent values.

	Property-based test pass over the display-options surface. New stream_data dependency and test/kino/ex_ratatui/property_test.exs (async: true) with six properties: every valid display value lands in the display map unchanged across all seven keys; the :dark / :light / :livebook atom shorthands round-trip; setting one display key never disturbs the other six (default fall-through invariant); reserved keys never appear in mount_opts after new/2 regardless of input shape; non-reserved keys are preserved in mount_opts in their original order; and frame/2 with any opt outside its supported set always raises ArgumentError. Two more properties in configure_test.exs (async: false) cover global-state behavior: configure/1 stores each {key, value} pair under Application.get_env(:kino_ex_ratatui, key), and per-instance opts always win over configure/1 for the same key (precedence invariant). Generators per key, uniq_list_of to keep keyword-key uniqueness deterministic, and explicit Application.delete_env cleanup between iterations so the global-state properties stay order-independent.

	:theme atom shorthands — :dark, :light, :livebook. In addition to a full xterm.js ITheme map, :theme now accepts three atoms resolved in the JS hook. :dark picks a bundled Catppuccin Mocha-flavored palette (the same colors as the no-opts default); :light picks Catppuccin Latte. :livebook reads the user's OS-level prefers-color-scheme and live-switches between :dark / :light whenever that preference changes — no cell re-eval needed. The previous map-only validator is unchanged; passing an unknown atom (e.g. theme: :neon) still raises ArgumentError with a message naming the offending option. JS-side: a subscribeLivebookTheme/2 helper on the live and static paths registers a matchMedia("(prefers-color-scheme: dark)") listener for the iframe's lifetime; iframes torn down on cell re-eval take the listener with them. Theming example extended with sections 4–5 walking through :livebook autoswitch and configure/1.

	Per-instance display options on new/2 and frame/2. Seven reserved opts now configure the xterm.js iframe per cell: :theme (full xterm.js ITheme map — background, foreground, cursor, cursorAccent, selectionBackground, the 16-color ANSI palette, etc.), :font_family (CSS string), :font_size (px integer), :height (CSS length applied to the xterm container — accepts "600px", "60vh", "calc(100vh - 200px)", …), :cursor_blink (boolean), :scrollback (non-negative integer), and :stopped_message (string painted into the iframe when the runtime exits via {:stop, _} or a mount failure). Defaults preserve the previous look exactly. Reserved keys are stripped from the keyword list before reaching ExRatatui.App.mount/1, so apps never see them as mount opts. Each value is validated at the call site — bad shapes raise ArgumentError with a message naming the offending option, never silently. frame/2 accepts the static-friendly subset (:theme, :font_family, :font_size); live-only opts and unknown keys raise so typos like col: instead of cols: no longer go through silently. The display payload flows to the JS bundle via handle_connect/1 (live mode) and the static info map (frame mode); the JS hook merges it onto its own copy of the defaults so out-of-band callers (custom payloads, future smart-cell variants) still get sensible values when only some opts are supplied. New examples/theming.livemd walks through every option — One Dark, Solarized Light, custom Fira Code at 16px, no-blink + custom stopped message, and a side-by-side static gallery via Kino.Layout.grid/1 of three theme variants on the same widget tree.

	Kino.ExRatatui.Telemetry — :telemetry integration. Mirrors the shape of ExRatatui.Telemetry one layer up, emitting events at the boundaries this widget controls so consumers can plug in logging, metrics, or distributed tracing without reaching into the runtime. Two span events ([:kino_ex_ratatui, :transport, :connect] with :mod/:width/:height — wraps the lazy Session + Transport.start_server/1 boot triggered by the first "resize"; [:kino_ex_ratatui, :render, :frame] with :mod/:byte_count — wraps the IO.iodata_to_binary/1 + Kino-bridge broadcast per-frame work) and three single events ([:kino_ex_ratatui, :transport, :disconnect] with :mod/:reason — fires exactly once per session, either from the runtime server's :DOWN or from the widget's terminate/2 if the runtime is still alive; [:kino_ex_ratatui, :input, :forward] with :mod/:byte_count — fires when bytes from xterm.js are forwarded to ByteStream.forward_input/3; [:kino_ex_ratatui, :resize] with :mod/:width/:height — fires on resizes after the boot one). Public helpers: span/3, execute/3, attach_default_logger/1, detach_default_logger/0. New Telemetry guide walks through the full event catalogue and a Telemetry.Metrics wiring example. Added {:telemetry, "~> 1.0"} as an explicit dependency and to extra_applications so the handler registry is available wherever the kino runs.


0.1.1 - 2026-04-30
Added
	README demo GIF (assets/demo.gif) showing a Kino.ExRatatui widget driving an ExRatatui.App inside a Livebook notebook.

0.1.0 - 2026-04-29
Added
	First release. kino_ex_ratatui runs an ExRatatui.App inside a Livebook notebook via xterm.js, implemented as a ~150-line Kino.JS.Live widget on top of ExRatatui.Transport.ByteStream. Two entry points: Kino.ExRatatui.new/2 for live App-driven kinos, Kino.ExRatatui.frame/2 for one-shot static frames suitable for docs and Kino.Layout.grid/1 side-by-side comparisons.
	JS bundle under assets/ — @xterm/xterm 5.5 + @xterm/addon-fit 0.10 bundled with esbuild 0.28 to lib/assets/kino_ex_ratatui/{main.js,main.css}. The bundle is committed so installing the hex package needs no Node toolchain. mix assets.install and mix assets.build aliases are provided for contributors.
	Lazy lifecycle. The runtime server and ExRatatui.Session are created on the first "resize" event from the iframe, so dimensions always come from xterm.js's FitAddon rather than a hardcoded default. Subsequent resize events flow through ByteStream.forward_resize/4. When the App returns {:stop, _} (or mount/1 fails), the widget broadcasts the canonical alt-screen leave sequence so xterm.js restores its cursor and main buffer.
	Test suite — 22 tests via Kino.Test's configure_livebook_bridge + push_event/3 + assert_broadcast_event/3, covering: lazy boot, mount-opts pass-through, handleconnect payload, first/subsequent resize, input round-trip, input arriving before first resize, server :DOWN, terminate cleanup, mount failure, unrelated handle_info messages, and `_assets_info/0. Runs async in 0.2s, 100% line coverage (test fixtures excluded viatest_coverage: [ignore_modules: [...]]`).
	Three bundled example notebooks under examples/ — system_monitor.livemd (callback-runtime dashboard porting ex_ratatui/examples/system_monitor.exs with Gauge, Table, /proc polling), chat_interface.livemd (callback-runtime AI-chat mock exercising Markdown, Textarea, Throbber, Scrollbar, and /-prefixed SlashCommands autocomplete via Popup — ported from the original imperative ExRatatui.run/1 loop in ex_ratatui/examples/chat_interface.exs), and reducer_counter.livemd (reducer-runtime counter with a Subscription.interval plus a Kino.ExRatatui.frame/2 static-frame demo). Each notebook cross-references the other two and links to the relevant runtime guide so any one of them is a complete jumping-off point.



  

    Configuration

kino_ex_ratatui reads display options from three sources, in order:
per-instance opts (new/2 / frame/2)  >  configure/1  >  module defaults
Each key is resolved independently, so a global font and theme can be set via configure/1 while still overriding just the theme on an individual cell.
Per-instance opts
The most direct path. See the Display options table for the full list.
Kino.ExRatatui.new(MyApp,
  theme: %{background: "#0d1117", foreground: "#c9d1d9"},
  font_size: 14,
  height: "600px"
)
Global defaults via configure/1
Kino.ExRatatui.configure/1 writes to the :kino_ex_ratatui Application environment and applies to every subsequent new/2 and frame/2 call.
# In a Livebook setup cell:
Kino.ExRatatui.configure(
  theme: :livebook,
  font_family: "JetBrains Mono, ui-monospace, monospace",
  font_size: 14
)
Calling configure/1 again merges into the existing config rather than replacing it — handy for splitting concerns across cells:
Kino.ExRatatui.configure(font_size: 14)         # font choice in one place
Kino.ExRatatui.configure(theme: :livebook)      # theme in another
# Both keys are now active.
The same values are available via Application.get_all_env(:kino_ex_ratatui), so a release config/runtime.exs or Config-driven release config works equivalently:
# config/runtime.exs
config :kino_ex_ratatui,
  theme: :livebook,
  font_size: 14
Validation runs immediately — configure/1 raises ArgumentError on bad shapes the same way new/2 does, so a mistyped key fails at the source rather than producing a working-but-wrong widget.
Atom theme shorthands
The :theme option accepts a full xterm.js ITheme map (the lowest-level form) or one of three atoms:
	Atom	What it does
	:dark	Bundled Catppuccin Mocha-flavored dark palette. Same colors as the no-opts default.
	:light	Bundled Catppuccin Latte-flavored light palette. Pairs visually with :dark.
	:livebook	Picks :dark or :light based on the user's OS-level prefers-color-scheme, and re-applies on the fly when that preference changes.

The atoms are resolved in the JS hook, so theme: :livebook is a single source of truth for "match my Livebook" — no duplicate light/dark configs needed.
# Reactively follows the user's color-scheme preference.
Kino.ExRatatui.configure(theme: :livebook)

# Or pick one explicitly.
Kino.ExRatatui.new(MyApp, theme: :light)
For finer control (selection backgrounds, the 16-color ANSI palette, cursor accents) supply a map directly — same vocabulary as xterm.js's Terminal({theme: ...}):
Kino.ExRatatui.new(MyApp,
  theme: %{
    background: "#282c34",
    foreground: "#abb2bf",
    cursorAccent: "#282c34",
    selectionBackground: "#3e4451",
    red: "#e06c75",
    green: "#98c379",
    # …
  }
)
Merge order in practice
Kino.ExRatatui.configure(theme: :livebook, font_size: 14, height: "600px")

# This call uses:
#   - theme: %{background: "#000"}    (per-instance — overrides configure's :livebook)
#   - font_size: 14                   (configure)
#   - height: "600px"                 (configure)
#   - cursor_blink: true              (module default — neither set it)
Kino.ExRatatui.new(MyApp, theme: %{background: "#000"})
Each key is resolved independently — there's no "all-or-nothing" inheritance.
When configure/1 is the wrong tool
If a TUI app needs runtime knowledge (current user, request id, …), thread that through ExRatatui.App.mount/1 instead. configure/1 is for cosmetics: theme, font, height, cursor, scrollback, stopped message — values that don't change between cells in the same notebook.


  

    Telemetry

kino_ex_ratatui emits :telemetry events at the boundaries the widget itself controls. They sit one layer above the events ex_ratatui already emits — together they give a complete profile of a Livebook-driven TUI.
Event tree at a glance
[:kino_ex_ratatui, :transport, :connect]    span    — widget boots Session + runtime
[:kino_ex_ratatui, :transport, :disconnect] single  — server :DOWN or widget terminate
[:kino_ex_ratatui, :render, :frame]         span    — ANSI broadcast over the Kino bridge
[:kino_ex_ratatui, :input, :forward]        single  — bytes from xterm.js → runtime
[:kino_ex_ratatui, :resize]                 single  — resizes after the boot one
Span events emit :start / :stop / :exception suffixes. Most handlers only attach to :stop for timing and :exception for failures.
See Kino.ExRatatui.Telemetry for the full metadata reference.
Quick start: log every event
# In a Livebook setup cell, or the application's start/2 callback:
Kino.ExRatatui.Telemetry.attach_default_logger(level: :info)
Detach with Kino.ExRatatui.Telemetry.detach_default_logger/0.
Wiring Telemetry.Metrics
If Telemetry.Metrics is already running (e.g. in a Phoenix LiveDashboard), add these alongside whatever ex_ratatui metrics matter:
defmodule MyApp.Telemetry do
  import Telemetry.Metrics

  def metrics do
    [
      # Time-to-first-frame on widget boot.
      summary("kino_ex_ratatui.transport.connect.stop.duration",
        unit: {:native, :millisecond}
      ),

      # How many sessions opened / closed.
      counter("kino_ex_ratatui.transport.disconnect"),

      # Per-frame ANSI broadcast cost (typically microseconds).
      summary("kino_ex_ratatui.render.frame.stop.duration",
        unit: {:native, :microsecond}
      ),

      # Input bytes flowing client → server.
      sum("kino_ex_ratatui.input.forward.byte_count"),

      # Resize storms (windows, splits, …).
      counter("kino_ex_ratatui.resize")
    ]
  end
end
Pairing with ex_ratatui's events
The two trees are complementary, not duplicative:
	Concern	Owned by
	mount/1 runtime, App handle_event/2, render command building	[:ex_ratatui, ...]
	Widget boot (network handshake + first Transport.start_server/1), ANSI broadcast over the Kino bridge, client input forwarding	[:kino_ex_ratatui, ...]

Attach to whichever is needed. A typical setup attaches [:ex_ratatui, :runtime, :event, :stop] for App-level latency and [:kino_ex_ratatui, :render, :frame, :stop] for the wire cost — together they show where time is going without instrumenting either layer manually.
Custom handlers
The public helpers Kino.ExRatatui.Telemetry.span/3 and Kino.ExRatatui.Telemetry.execute/3 are thin wrappers around :telemetry.span/3 and :telemetry.execute/3 that prepend :kino_ex_ratatui to the event name. Use them when building a higher-level wrapper around Kino.ExRatatui.new/2 to emit events under the same namespace.
For one-off handlers, attach directly:
:telemetry.attach(
  "my-app-frame-tracker",
  [:kino_ex_ratatui, :render, :frame, :stop],
  &MyApp.TuiTracker.handle_frame/4,
  nil
)
Always use a captured module function (&MyApp.TuiTracker.handle_frame/4), not an anonymous function — :telemetry logs a performance warning otherwise.

Run an ExRatatui.App inside a Livebook notebook via xterm.js.
Kino.ExRatatui is a byte-stream transport that pipes the runtime
server's rendered ANSI through an xterm.js iframe and forwards
keypresses + resize events back.
Example
defmodule Counter do
  use ExRatatui.App
  alias ExRatatui.Event.Key
  alias ExRatatui.Layout.Rect
  alias ExRatatui.Widgets.Paragraph

  def mount(_), do: {:ok, %{n: 0}}

  def render(state, frame) do
    [{%Paragraph{text: "Count: #{state.n}"},
      %Rect{x: 0, y: 0, width: frame.width, height: frame.height}}]
  end

  def handle_event(%Key{code: "+"}, s), do: {:noreply, %{s | n: s.n + 1}}
  def handle_event(%Key{code: "q"}, s), do: {:stop, s}
  def handle_event(_, s),                do: {:noreply, s}
end

Kino.ExRatatui.new(Counter)
Mount options
The second argument to new/2 is a keyword list. Any key not listed
under Display options below is forwarded
verbatim to ExRatatui.App.mount/1. Use it for per-instance
configuration the App reads from its mount opts.
The keys :mod, :name, and :transport are reserved by the
runtime and silently overwritten.
Display options
Reserved opts that configure the xterm.js iframe rather than the App.
All are optional; defaults preserve the current widget look. Unknown
values raise ArgumentError at the call site.
	Option	Type	Default	Notes
	:theme	map() or :dark / :light / :livebook	catppuccin-style dark theme	A map is forwarded to xterm.js's Terminal({theme: ...}) verbatim — accepts the full xterm.js ITheme object (:background, :foreground, :cursor, :cursorAccent, :selectionBackground, the 16-color ANSI palette, …). Atom keys are JSON-encoded as strings; use the camelCase xterm.js expects (cursorAccent, not cursor_accent). The atom shorthands resolve in the JS hook: :dark and :light pick a bundled palette, :livebook follows the user's prefers-color-scheme and live-switches when it changes.
	:font_family	String.t()	"ui-monospace, SFMono-Regular, Menlo, Monaco, Consolas, monospace"	CSS font-family value. Anything xterm.js can render.
	:font_size	pos_integer()	13	Cell font size in px.
	:height	String.t()	"400px"	CSS height applied to the xterm container. Accepts any valid CSS length ("600px", "60vh", …).
	:cursor_blink	boolean()	true	xterm.js cursor blink.
	:scrollback	non_neg_integer()	1000	xterm.js scrollback line limit.
	:stopped_message	String.t()	"App stopped — re-evaluate the cell to start a new run."	Shown in the iframe after the runtime exits ({:stop, _} or mount/1 failure).

Example:
Kino.ExRatatui.new(Counter,
  # mount opts → App.mount/1
  start: 10,
  # display opts → xterm.js
  theme: %{background: "#0d1117", foreground: "#c9d1d9"},
  font_size: 14,
  height: "600px"
)
Global configuration
Set defaults that apply to every new/2 and frame/2 call in the
current Livebook (or the application's runtime) with configure/1:
# In a setup cell, or the application's start/2:
Kino.ExRatatui.configure(
  theme: :livebook,
  font_family: "JetBrains Mono, ui-monospace, monospace",
  font_size: 14
)
Merge order is per-instance opts > configure/1 > module
defaults, key by key. Setting theme: globally and then passing
theme: %{background: "#000"} per-instance overrides the theme
alone — the global font_size still wins for that call. See
configure/1 for the full list and the
Configuration guide for recipes.
Lifecycle
Each new/2 call spawns a fresh Kino.JS.Live server. The runtime
server and ExRatatui.Session are created lazily on the first
"resize" event from the iframe so we always start at the correct
cell dimensions reported by xterm.js's FitAddon. Re-evaluating the
cell (or closing the notebook) tears the runtime server down and
starts a new one — no state is preserved across re-evals.
How it plugs into ExRatatui
Implements ExRatatui.Transport as a byte-stream transport, using
ExRatatui.Transport.start_server/1 to boot the runtime and
ExRatatui.Transport.ByteStream to pump input + resize events. See
the Custom Transports
guide for the
reference shape.
Inline images
The bundled JS hook loads @xterm/addon-image, so
ExRatatui.Widgets.Image renders end-to-end in Livebook over Sixel
and iTerm2 inline-image protocols. Build images with
ExRatatui.Image.new/2 and place them in the widget tree like any
other widget — no kino-side configuration is needed. The protocol is
picked explicitly at construction time; pass :sixel or :iterm2
(xterm.js does not implement the Kitty graphics protocol). See the
Images guide for the
full API.

      


      
        Summary


  
    Functions
  


    
      
        configure(opts)

      


        Sets global defaults applied to every subsequent new/2 and
frame/2 call in the current runtime.



    


    
      
        frame(widgets, opts \\ [])

      


        Renders a one-shot static frame of widgets and returns a
non-interactive Kino.JS widget that paints it once.



    


    
      
        new(mod, opts \\ [])

      


        Builds a new live kino that hosts mod (an ExRatatui.App).

      configure(opts)



        
          
        

    

  


  

      

          @spec configure(keyword()) :: :ok


      


Sets global defaults applied to every subsequent new/2 and
frame/2 call in the current runtime.
Accepts the same keys as the Display options
on new/2. Each value is validated immediately (the same way it
would be on new/2) — bad shapes raise ArgumentError.
Per-instance opts on new/2 / frame/2 still win key-by-key.
Calling configure/1 again merges into the existing config rather
than replacing it, so related settings can be split across cells.
Returns :ok.
Examples
# Reactively follow Livebook's light/dark mode and bump the font.
Kino.ExRatatui.configure(theme: :livebook, font_size: 14)

# Per-instance overrides still apply. This call uses the configured
# font_size: 14 but a custom theme.
Kino.ExRatatui.new(Counter, theme: %{background: "#000"})
Stored under the :kino_ex_ratatui Application environment, so the
same value is also reachable via Application.get_all_env(:kino_ex_ratatui)
when orchestrating from a release config/runtime.exs or any
other Config-driven setup.

  



    

  
    
      
    
    
      frame(widgets, opts \\ [])



        
          
        

    

  


  

      

          @spec frame(
  [{ExRatatui.widget(), ExRatatui.Layout.Rect.t()}],
  keyword()
) :: Kino.JS.t()


      


Renders a one-shot static frame of widgets and returns a
non-interactive Kino.JS widget that paints it once.
Useful for documentation, screenshots in notebooks, or
Kino.Layout.grid([frame_a, frame_b, frame_c]) side-by-side
comparisons. There is no event loop, no resize handling, and no
runtime server — just an ExRatatui.Session rendered once and
written to xterm.js.
Options
	:cols — terminal width in cells. Defaults to 80.
	:rows — terminal height in cells. Defaults to 24.
	:theme — see Display options. Same
defaults as new/2.
	:font_family — same.
	:font_size — same.

Live-only display opts (:height, :cursor_blink, :scrollback,
:stopped_message) are not accepted here and will raise
ArgumentError — they have nothing to apply against in a static
one-shot frame.
Examples
iex> alias ExRatatui.Layout.Rect
iex> alias ExRatatui.Widgets.{Block, Paragraph}
iex> kino = Kino.ExRatatui.frame(
...>   [
...>     {%Paragraph{
...>        text: "Hello from a static frame!",
...>        block: %Block{title: "demo"}
...>      },
...>      %Rect{x: 0, y: 0, width: 40, height: 5}}
...>   ],
...>   cols: 40,
...>   rows: 5
...> )
iex> kino.module
Kino.ExRatatui

  



    

  
    
      
    
    
      new(mod, opts \\ [])



        
          
        

    

  


  

      

          @spec new(
  module(),
  keyword()
) :: Kino.JS.Live.t()


      


Builds a new live kino that hosts mod (an ExRatatui.App).
opts is split into Display options
consumed by the widget itself and Mount options
forwarded verbatim to ExRatatui.App.mount/1. Reserved display
keys never reach the App.
Examples
# Plain
Kino.ExRatatui.new(Counter)

# Mount opt forwarded to mount/1
Kino.ExRatatui.new(Counter, start: 10)

# Display opt + mount opt
Kino.ExRatatui.new(Counter,
  start: 10,
  theme: %{background: "#0d1117", foreground: "#c9d1d9"},
  font_size: 14
)

:telemetry integration for kino_ex_ratatui.
Mirrors the shape of ExRatatui.Telemetry
one layer up — events fire at the boundaries this package controls
(Kino widget boot, ANSI broadcast, client input forward, widget
shutdown) rather than at the runtime / session layer ex_ratatui
already instruments. Both fire concurrently when a Livebook-driven
TUI is running, and consumers attach handlers to whichever they
need.
Why a separate event tree
ex_ratatui measures the App / runtime layer:
[:ex_ratatui, :runtime, :event], [:ex_ratatui, :render, :frame],
[:ex_ratatui, :session, :lifecycle, *], and so on. Those events
fire from inside the ex_ratatui runtime server regardless of transport.
kino_ex_ratatui adds events for the Kino-side cost:
	widget boot (lazy Session + runtime construction triggered by
the first "resize" from xterm.js)
	the per-frame work the widget itself does on top of the
runtime: turning the runtime's iodata into a binary and pushing
it over the Kino bridge to xterm.js
	client input being forwarded from the iframe to the runtime
	subsequent resize forwards (the first resize is rolled into the
:transport, :connect span)
	widget teardown

Attaching handlers to BOTH event trees gives a complete profile of
a Livebook-driven TUI without double-counting any single operation.
Events
All events are prefixed with :kino_ex_ratatui.
Span events (:start / :stop / :exception)
Each span emits three events with the suffix appended. Handlers
typically attach to :stop for timing and :exception for
failure tracking.
	Event	Description	Metadata
	[:kino_ex_ratatui, :transport, :connect]	First "resize" event boots ExRatatui.Session + runtime server (mount/1 runs inside the span).	:mod, :width, :height
	[:kino_ex_ratatui, :render, :frame]	Per-frame Kino-side work: IO.iodata_to_binary/1 + the broadcast_event/3 helper from Kino.JS.Live over the bridge.	:mod, :byte_count

:start events carry %{monotonic_time: integer, system_time: integer}
as measurements. :stop events add :duration (native units). On
exception the metadata gains :kind, :reason, and :stacktrace.
Single events
	Event	Description	Measurements	Metadata
	[:kino_ex_ratatui, :transport, :disconnect]	Runtime server exited ({:DOWN, _, _, _, _}) or the Kino widget is terminating. Fires exactly once per session — when :DOWN clears the server refs, the widget's terminate/2 becomes a no-op.	%{system_time: integer}	:mod, :reason
	[:kino_ex_ratatui, :input, :forward]	A keypress (or other byte payload) from xterm.js was forwarded to the runtime via ExRatatui.Transport.ByteStream.forward_input/3.	%{system_time: integer}	:mod, :byte_count
	[:kino_ex_ratatui, :resize]	A resize after the initial boot was forwarded to the runtime via ExRatatui.Transport.ByteStream.forward_resize/4.	%{system_time: integer}	:mod, :width, :height

Attaching a default logger
# In a Livebook setup cell (or the application's start/2):
Kino.ExRatatui.Telemetry.attach_default_logger()
That attaches a handler logging every :stop and single event at
:debug level. Pass attach_default_logger(level: :info) to bump
the level. Detach with detach_default_logger/0.
Real apps wire Telemetry.Metrics instead:
defmodule MyApp.Telemetry do
  import Telemetry.Metrics

  def metrics do
    [
      summary("kino_ex_ratatui.transport.connect.stop.duration",
        unit: {:native, :millisecond}),
      counter("kino_ex_ratatui.transport.disconnect"),
      summary("kino_ex_ratatui.render.frame.stop.duration",
        unit: {:native, :microsecond}),
      counter("kino_ex_ratatui.input.forward")
    ]
  end
end

    Functions
  


    
      
        attach_default_logger(opts \\ [])

      


        Attaches a logger that prints every kino_ex_ratatui telemetry
event. Useful during development; detach with
detach_default_logger/0.



    


    
      
        detach_default_logger()

      


        Detaches the default logger previously attached with
attach_default_logger/1.



    


    
      
        execute(event, measurements, meta)

      


        Emits a single :telemetry event rooted at
[:kino_ex_ratatui | event].



    


    
      
        span(event, meta, fun)

      


        Wraps fun in a :telemetry span rooted at
[:kino_ex_ratatui | event].



    





      


      
        Functions


        


    

  
    
      
    
    
      attach_default_logger(opts \\ [])



        
          
        

    

  


  

      

          @spec attach_default_logger(keyword()) :: :ok | {:error, :already_exists}


      


Attaches a logger that prints every kino_ex_ratatui telemetry
event. Useful during development; detach with
detach_default_logger/0.
Options
	:level — log level (default: :debug).
	:events — list of event suffixes to attach (default: all
:stop and single events).


  



  
    
      
    
    
      detach_default_logger()



        
          
        

    

  


  

      

          @spec detach_default_logger() :: :ok | {:error, :not_found}


      


Detaches the default logger previously attached with
attach_default_logger/1.

  



  
    
      
    
    
      execute(event, measurements, meta)



        
          
        

    

  


  

      

          @spec execute([atom(), ...], map(), map()) :: :ok


      


Emits a single :telemetry event rooted at
[:kino_ex_ratatui | event].
:system_time is added to the measurements automatically if not
already present.

  



  
    
      
    
    
      span(event, meta, fun)



        
          
        

    

  


  

      

          @spec span([atom(), ...], map(), (-> term())) :: term()


      


Wraps fun in a :telemetry span rooted at
[:kino_ex_ratatui | event].
The fun's return value is returned unchanged. The given meta
is forwarded to both the :start and :stop events.

